

```

MODEL:
! Assign items with given weights to bundles so the
  total weights of each bundle are close to identical ;

! Keywords: Assignment, Bundling, Knapsack Model, LINGO;

SETS:
  ITEM: IWGT;
  BUNDLE: BWGT;
  IxB( ITEM, BUNDLE): Z;
ENDSETS

DATA:
! Assign these 13 weights to 4 bundles of close to equal size;
IWGT = 1260 3295 5302 5315 1579 3165 2513 2845 1843 1949 3102 2942 2513;
BUNDLE = 1..4;

! In the above we read the inputs in-place. There are 5 ways of exchanging
data with LINGO:
  1) In-place, as we do above,
  2) With Excel, using statements like X = @OLE() and @OLE() = Y.
     The LINGO model can also be stored in an Excel spreadsheet.
  3) With text files, using statements like X= @FILE() and @WRITE( Y).
  4) With an ODBC compatible database, using statements like X=@ODBC() and @ODBC() = Y.
     Some compatible databases include Access, Oracle and SQL.
  5) Via a calling application's memory locations, using statements like
     X = @POINTER( 1), and @POINTER( 2) = Y.
;
ENDDATA

SUBMODEL Bundlem:
! Minimize difference between the smallest and largest bundle;
MIN = BWX - BWN;
! Each item must be assigned to a bundle;
@FOR( ITEM( i):
  @SUM( BUNDLE( b): Z( i, b)) = 1;
);

! Compute bundle weights for each BUNDLE b;
@FOR( BUNDLE( b):
  BWGT( b) = @SUM( ITEM( i): IWGT( i)* Z( i, b));
);

! Enforce max and min weight value for each BUNDLE b;
@FOR( BUNDLE( b):
  BWX >= BWGT( b);
  BWN <= BWGT( b);
);

! The Z( i, b) must be 0 or 1 ( Binary) for each
  ITEM and BUNDLE pair in the set IxB;
@FOR( IxB( i, b): @BIN( Z( i, b)));
ENDSUBMODEL

CALC:
@SET( 'OROUTE',1); ! Route output to Lingo's command window;
@SET( 'TERSEO',2); ! Output level (0:verbose, 1:terse, 2:only errors, 3:none);
@SOLVE( Bundlem); ! Solve a specified model. There may be > 1;

@WRITE( ' Assign Items to Bundles So Wgts Are Close To Equal', @NEWLINE( 1));
@WRITE( ' Bundle      Weight', @NEWLINE( 1));

@FOR( BUNDLE( b):
  @WRITE( '      ', BUNDLE( b), '      ', BWGT( b), @NEWLINE( 1));
);

@WRITE( @NEWLINE( 1), ' Bundle      Item      Weight', @NEWLINE( 1));
@FOR( BUNDLE( b):

```

```
@FOR( ITEM( i) | Z( i, b) #GT# 0.5:
  @WRITE( '      ', BUNDLE( b), '      ', @FORMAT( ITEM( i), ' 6s'), '      ', IWGT( i),
@NEWLINE( 1));
  ); );
ENDCALC
END
```